

9 Proof by induction

The laws we use in equational reasoning about programs have to be justified. Because we are dealing with lists, which are a recursive data type, functions on lists are often defined by recursion, so proofs about them they will usually be justified by induction.

Think about

```
> pow :: Num a => a -> Int -> a
> pow x 0 = 1
> pow x n = pow x (n-1) * x
```

How would we prove

$$\text{pow } x \ (m + n) = \text{pow } x \ m \times \text{pow } x \ n$$

Being honest we would really want the n to be a natural number, and

```
> pow x 0 = 1
> pow x n | n > 0 = pow x (n-1) * x
```

is closer to it. In writing mathematics, however, one would be more likely to say

$$\begin{aligned} \text{pow } x \ 0 &= 1 \\ \text{pow } x \ (n+1) &= \text{pow } x \ n * x \end{aligned}$$

with the observation that since $n \in \mathbb{N}$, the $n + 1$ cannot be zero. Older textbooks may have Haskell definitions that look like this, but the syntactic form was misleading and has been removed from the language.

9.1 Induction over natural numbers

A proof of $P(n)$ for every natural number n is an infinite number of proofs, one for $P(0)$, one for $P(1)$, one for $P(2)$, ... Clearly providing all of those will take a long time. A better alternative would be to provide a systematic way of generating any one of those proofs.

Every natural number is either 0, or is $n + 1$ for some natural number n . To prove $P(n)$ for every natural number n it is enough to prove

1. $P(0)$
2. for every natural number n , if $P(n)$ then $P(n + 1)$.

Since any natural number n can be made of a 0 and some number of applications of $(+1)$, the proof of $P(n)$ can be built out of a proof of $P(0)$ and some (the same) number of proofs of $P(0) \Rightarrow P(1)$, $P(1) \Rightarrow P(2)$, $\dots P(n-1) \Rightarrow P(n)$. So a proof of

$$pow\ x\ (m + n) = pow\ x\ m \times pow\ x\ n$$

by induction on n would consist to two parts: one for $n = 0$,

$$\begin{aligned} & pow\ x\ (m + 0) \\ = & \{ \text{unit of addition} \} \\ & pow\ x\ m \\ = & \{ \text{unit of multiplication} \} \\ & pow\ x\ m \times 1 \\ = & \{ \text{definition of } pow \} \\ & pow\ x\ m \times pow\ x\ 0 \end{aligned}$$

and assuming, temporarily, $pow\ x\ (m + n) = pow\ x\ m \times pow\ x\ n$ for some n ,

$$\begin{aligned} & pow\ x\ (m + (n + 1)) \\ = & \{ \text{definition of } (+) \text{ (perhaps associativity of } (+)) \} \\ & pow\ x\ ((m + n) + 1) \\ = & \{ \text{definition of } pow \} \\ & pow\ x\ (m + n) \times x \\ = & \{ \text{induction hypothesis} \} \\ & (pow\ x\ m \times pow\ x\ n) \times x \\ = & \{ \text{associativity of multiplication} \} \\ & pow\ x\ m \times (pow\ x\ n \times x) \\ = & \{ \text{definition of } pow \} \\ & pow\ x\ m \times pow\ x\ (n + 1) \end{aligned}$$

Then we can discharge the assumption and read this as a proof of

$$\begin{aligned} & \text{if } pow\ x\ (m + n) = pow\ x\ m \times pow\ x\ n \text{ then} \\ & \quad pow\ x\ (m + (n + 1)) = pow\ x\ m \times pow\ x\ (n + 1) \end{aligned}$$

for every natural number n .

9.2 The take lemma

Sometimes proofs about lists can be reduced to proofs about natural numbers. The *take lemma* is the observation that two lists xs and ys are equal provided

that $\text{take } n \text{ } xs = \text{take } n \text{ } ys$ for all natural numbers n . That requirement can often be proved by induction on n , noting that the case $n = 0$ always holds because both sides are null.

For example, $\text{map } f (\text{iterate } f \text{ } x) = \text{iterate } f (f \text{ } x)$ because

$$\begin{aligned}
 & \text{take } (n + 1) (\text{map } f (\text{iterate } f \text{ } x)) \\
 = & \{ \text{definition of } \text{iterate} \} \\
 & \text{take } (n + 1) (\text{map } f (x : \text{iterate } f (f \text{ } x))) \\
 = & \{ \text{definition of } \text{map} \} \\
 & \text{take } (n + 1) (f \text{ } x : \text{map } f (\text{iterate } f (f \text{ } x))) \\
 = & \{ \text{definition of } \text{take} \} \\
 & f \text{ } x : \text{take } n (\text{map } f (\text{iterate } f (f \text{ } x))) \\
 = & \{ \text{induction hypothesis} \} \\
 & f \text{ } x : \text{take } n (\text{iterate } f (f (f \text{ } x))) \\
 = & \{ \text{definition of } \text{take} \} \\
 & \text{take } (n + 1) (f \text{ } x : \text{iterate } f (f (f \text{ } x))) \\
 = & \{ \text{definition of } \text{iterate} \} \\
 & \text{take } (n + 1) (\text{iterate } f (f \text{ } x))
 \end{aligned}$$

Notice that a proof about $n + 1$ and x requires a hypothesis about n and $f \text{ } x$. Formally, the induction hypothesis is that *for all* x the result holds for n .

It should be clear that by induction on n , the take lemma shows that each corresponding pair of finite prefixes of xs and ys is equal. Informally, it is clear that this means that xs and ys must be equal even if infinite. We will return to the reason for this later.

An equivalent presentation of the take lemma uses the function

```

> take' n [] | n > 0 = []
> take' n (x:xs) | n > 0 = x : take' (n-1) xs

```

which prunes its list argument to no more than n constructors, but leaves \perp values after that. It can be generalised to similar proof schemes for other data types, for example

```

> data Stream a = Cons a (Stream a)

```

the type of infinite streams which have no end, or types where there are a several possible last constructors.

9.3 Induction over finite lists

Every finite list is either $[]$, or is $x : xs$ for some finite list xs . To prove $P(xs)$ for every finite list xs it is enough to prove

1. $P([])$
2. for every finite list xs , if $P(xs)$ then $P(x : xs)$.

Recall that

```
> (++) :: [a] -> [a] -> [a]
> []      ++ ys = ys
> (x:xs) ++ ys = x:(xs++ys)
```

We prove that $(xs ++ ys) ++ zs = xs ++ (ys ++ zs)$ for all finite lists xs by induction on xs . Firstly for empty lists

```
([] ++ ys) ++ zs
= { definition of (++) }
  ys ++ zs
= { definition of (++) }
  [] ++ (ys ++ zs)
```

then assuming $(xs ++ ys) ++ zs = xs ++ (ys ++ zs)$

```
((x : xs) ++ ys) ++ zs
= { definition of (++) }
  (x : (xs ++ ys)) ++ zs
= { definition of (++) }
  x : ((xs ++ ys) ++ zs)
= { induction hypothesis }
  x : (xs ++ (ys ++ zs))
= { definition of (++) }
  (x : xs) ++ (ys ++ zs)
```

so the result is true for all finite lists xs (and all lists ys and zs).

9.4 A second example

Given the definition

```
> reverse [] = []
> reverse (x:xs) = reverse xs ++ [x]
```

a proof of $\text{reverse} (\text{reverse } xs) = xs$ for all finite lists xs proceeds by induction on xs .

$$\begin{aligned} & \text{reverse} (\text{reverse } []) \\ = & \{ \text{definition of } \text{reverse} \} \\ & \text{reverse } [] \\ = & \{ \text{definition of } \text{reverse} \} \\ & [] \end{aligned}$$

and then assuming $\text{reverse} (\text{reverse } xs) = xs$ for some xs

$$\begin{aligned} & \text{reverse} (\text{reverse } (x : xs)) \\ = & \{ \text{definition of } \text{reverse} \} \\ & \text{reverse} (\text{reverse } xs ++ [x]) \\ = & \{ \text{unjustified step} \} \\ & x : \text{reverse} (\text{reverse } xs) \\ = & \{ \text{inductive hypothesis} \} \\ & x : xs \end{aligned}$$

The (as yet) unjustified step requires a lemma:

$$\text{reverse} (ys ++ [x]) = x : \text{reverse } ys$$

which can be proved by induction on ys .

$$\begin{aligned} & \text{reverse} ([] ++ [x]) \\ = & \{ \text{definition of } (++) \} \\ & \text{reverse } [x] \\ = & \{ \text{definition of } \text{reverse} \} \\ & \text{reverse } [] ++ [x] \\ = & \{ \text{definition of } \text{reverse} \} \\ & [] ++ [x] \\ = & \{ \text{definition of } (++) \} \\ & [x] \\ = & \{ \text{definition of } \text{reverse} \} \\ & x : \text{reverse } [] \end{aligned}$$

and assuming it to be true for ys

$$\begin{aligned}
 & \text{reverse } ((y : ys) ++ [x]) \\
 = & \{ \text{definition of } (++) \} \\
 & \text{reverse } (y : (ys ++ [x])) \\
 = & \{ \text{definition of reverse} \} \\
 & \text{reverse } (ys ++ [x]) ++ [y] \\
 = & \{ \text{inductive hypothesis} \} \\
 & (x : \text{reverse } ys) ++ [y] \\
 = & \{ \text{definition of } (++) \} \\
 & x : (\text{reverse } ys ++ [y]) \\
 = & \{ \text{definition of reverse} \} \\
 & x : \text{reverse } (y : ys)
 \end{aligned}$$

9.5 Induction over partial lists

A partial list is one with a tail that is either \perp or a smaller partial list. There is a similar principle of induction over partial lists; to prove $P(xs)$ for every partial list xs it is enough to prove

1. $P(\perp)$
2. for every partial list xs , if $P(xs)$ then $P(x : xs)$.

For example $xs ++ ys = xs$ for all partial lists xs .

The case $\perp ++ ys = \perp$ is immediate from the strictness of $(++)$, since it is defined by pattern matching on its left argument. Then assuming the result for some partial xs

$$\begin{aligned}
 & (x : xs) ++ ys \\
 = & \{ \text{definition of } (++) \} \\
 & x : (xs ++ ys) \\
 = & \{ \text{inductive hypothesis} \} \\
 & x : xs
 \end{aligned}$$

9.6 Infinite lists

An infinite list can be thought of as a limit of a sequence of partial lists, for example $[0..]$ is the limit of the chain $\perp, 0 : \perp, 0 : 1 : \perp, 0 : 1 : 2 : \perp, \dots$ and so on. Successive elements of this chain are related by the information ordering, (\sqsubseteq) . This is the partial order that has $\perp \sqsubseteq x$ for all x , and $(x : xs) \sqsubseteq (y : ys)$ whenever $x \sqsubseteq y$ and $xs \sqsubseteq ys$.

A property P is called chain complete if whenever $P(xs_i)$ holds for every element xs_i of a chain it also holds for the limit of that chain. So, chain complete properties that also hold for all partial lists will be ones that also hold for infinite lists.

It is enough to know that positive properties are chain complete. These include universally quantified equations between Haskell-definable expressions, and conjunctions ('and's) of positive properties. Inequalities need not be chain complete, nor need properties involving existential quantification be. For example " $drop\ n\ xs = \perp$ for some n " is true for all partial lists, but is not true for infinite lists. It is not one equation, but the disjunction ('or') of an infinite number of equations.

The earlier proof of $(xs ++ ys) ++ zs = xs ++ (ys ++ zs)$ for finite lists can be extended to partial lists xs , because $(\perp ++ ys) ++ zs = \perp ++ zs = \perp$ (by strictness of $(++)$) so the result holds for all finite and partial lists, and since it is an equation between Haskell expressions it is chain complete and also holds for infinite lists. (Informally, both sides are equal to xs if xs is an infinite list.)

Can we extend the proof of $reverse\ (reverse\ xs) = xs$ to infinite lists? Certainly it holds for bottom: $reverse\ (reverse\ \perp) = reverse\ \perp = \perp$ (again, by strictness of $reverse$). This looks promising; but running $reverse\ (reverse\ [0..])$ will produce no output: $reverse\ (reverse\ xs) = \perp$ for all partial xs .

What went wrong? The proof that

$$reverse\ (reverse\ xs) = xs \Rightarrow reverse\ (reverse\ (x : xs)) = x : xs$$

required a lemma $reverse\ (ys ++ [x]) = x : reverse\ ys$ which we only proved for finite lists ys . It is not true for infinite lists, nor indeed for partial lists, because $reverse\ (\perp ++ [x]) = \perp$ and so $reverse\ (ys ++ [x]) = \perp$ for all partial lists ys .

9.7 Summary: induction schemes

A proof by (structural) induction on xs that a property $P(xs)$ holds for all of some class of lists xs involves assembling proof components for all of the ways of constructing list in that class.

To prove that $P(xs)$ for all finite xs , it is enough to prove (i) $P([])$ and (ii) if $P(xs)$ for some finite xs then $P(x : xs)$.

To prove that $P(xs)$ for all partial xs , it is enough to prove (i) $P(\perp)$ and (ii) if $P(xs)$ for some partial xs then $P(x : xs)$.

To prove that $P(xs)$ for all infinite xs it is enough to prove (i) $P(\perp)$ and (ii) if $P(xs)$ for some partial xs then $P(x : xs)$, and (iii) P is chain complete (for example, because it is an equation between Haskell-definable expressions).

In the special case of an equation $E = F$ it is also possible to use the take lemma which requires that you can prove only that if $take\ n\ E = take\ n\ F$ then $take\ (n + 1)\ E = take\ (n + 1)\ F$.

9.8 Aside: Continuity (beyond the scope of this course)

This use of the words *limit* and *continuity* turn out to be exactly the same as you might meet in analysis.

You can define a distance function $d :: ([a], [a]) \rightarrow \mathbb{R}_{\geq 0}$, a *metric*, which behaves like the distance between points in space. In particular, the triangle inequality $d(xs, ys) + d(ys, zs) \geq d(xs, zs)$ holds.

A suitable distance between distinct lists xs and ys might be $\frac{1}{n}$ for the biggest n for which the *take' n xs = take' n ys* (or the equivalent equation for the standard *take* function). The successive elements of the sequence of partial approximations to an infinite list are arbitrarily close together, just like Cauchy sequences in \mathbb{R} . The infinite list is the limit of that Cauchy sequence.

Continuity for real valued functions turns out to be equivalent to preserving limits. A function has an abrupt discontinuity exactly when you can find a convergent sequence leading up to a point of discontinuity, so that the image of the limit under the function is not the limit of the images of the elements of the sequence.

Computable functions turn out to have to be not only monotonic (put in more information into an argument, they cannot retract any information that has already come out in the result). More than that they need to be continuous (apply a continuous function to a limit of a chain, and you get back the limit of the images of the sequence).

Monotonicity tells you that none of the information in the image of the limit of a sequence can contradict anything you might have learned from the image of an element of that sequence. Continuity tells you that all of the information in the image of the limit of the sequence will already have come out in the image of one of the elements of the sequence.

That is, if f is a computable function, and xs is an infinite list, then it is OK to compute $f\ xs$ by successively working out what you can know about $f\ xs$ by calculating $f\ (take'\ n\ xs)$ for successively bigger n .

Chain completeness also turns out to be continuity, and continuity guarantees soundness. A proof can be thought of as a function returning one of two values, with *Invalid* \sqsubseteq *Valid*. (You can think of it as *False* \sqsubseteq *True*, but do not confuse it with the usual information ordering on *Bool*.) A proof which is valid for all (except perhaps a finite number) of the elements of a chain cannot then make an abrupt change from valid to invalid at the infinite limit of the chain. Mathematical equality is continuous, so a mathematical equality of two Haskell expressions is the composition of continuous functions.

Exercises

9.1 Suppose a type of natural numbers is defined by

```
> data Nat = Zero | Succ Nat
```

Use recursion to define functions $int :: Nat \rightarrow Int$ and $nat :: Int \rightarrow Nat$ which embed the natural numbers in Int in the obvious way.

Use recursion (on the second argument) to define functions

```
add, mul, pow, tet :: Nat \rightarrow Nat \rightarrow Nat
```

which implement addition, multiplication, exponentiation, and what Goodstein calls tetration.

(x 'tet' $n = x \wedge x \wedge \dots \wedge x$ where there are n copies of x .)

9.2 What property characterises $foldNat$, the fold for Nat ? Define $foldNat$.

What are the deconstructors for Nat , and what characterises the unfold $unfoldNat$? Define $unfoldNat$.

Express each of int and nat as either $foldNat$ or $unfoldNat$.

Finally express add , mul , pow and tet as folds.

9.3 By definition, a *metric* on a set S is a function $d : S \times S \rightarrow \mathbb{R}$ which for all $x, y, z \in S$ satisfies

1. $d(x, y) \geq 0$
2. $d(x, y) = 0 \Leftrightarrow x = y$
3. $d(x, y) = d(y, x)$
4. $d(x, z) \leq d(x, y) + d(y, z)$

Show that d is a metric on lists (all of the same type $[\alpha]$ for some α) if $d(xs, ys) = 1/(k+1)$ where k is the smallest index for which $take\ k\ xs \neq take\ k\ ys$, if there is one; and $d(xs, ys) = 0$ if there is no such k .

9.4 A *Cauchy sequence* is, by definition, an infinite sequence whose elements get arbitrarily close together, and is expected therefore to have a limit. Specifically a sequence x is Cauchy if

$$\forall \varepsilon > 0. \exists n. \forall m \geq n. d(x_n, x_m) < \varepsilon.$$

Informally, which sequences of lists are Cauchy sequences with respect to the list metric in exercise 9.3? Describe the limit of such a sequence.

9.5 Provided that

```
p n    = True
p (c x y) = False
h (c x y) = x
t (c x y) = y
```

prove the claim on page 41 that

$$\mathit{unfold} \, p \, h \, t \cdot \mathit{fold} \, c \, n \; = \; \mathit{id}$$